Solutions 3

P3.1. Let $I := \langle xy - wz, wy - z^2, w^2 - xz \rangle$ be an ideal in the polynomial ring $\mathbb{Q}[w,x,y,z]$.

- **i.** Find (without using computer software) the reduced Gröbner basis of I with respect to the graded reverse lexicographic order and $w > x > y > z$.
- **ii.** Find (without using computer software) the reduced Gröbner basis of I with respect to the lexicographic order and $w > x > y > z$.
- **iii.** (*Bonus*) The ideal I has 8 distinct leading term ideals. Can you exhibit these eight monomial ideals?

Solution.

i. Set $f_1 := \underline{w\,y} - z^2$, $f_2 := \underline{x\,y} - w\,z$ and $f_3 := \underline{w^2} - x\,z$. Since

$$\begin{aligned}
\operatorname{spoly}(f_1, f_2) &= x\,f_1 - w\,f_2 = \underline{w^2\,z} - x\,z^2 = z\,f_3 \\
\operatorname{spoly}(f_1, f_3) &= w\,f_1 - y\,f_3 = \underline{x\,y\,z} - w\,z^2 = z\,f_1 \\
\operatorname{spoly}(f_2, f_3) &= w^2\,f_2 - x\,y\,f_3 = -\underline{w^3\,z} + x^2\,y\,z = -w\,z\,f_3 + x\,z\,f_2\,,
\end{aligned}$$

the Buchberger criterion shows that f_1, f_2, f_3 is the reduced Gröbner basis of I with respect to the graded reverse lexicographic order where $w > x > y > z$.

ii. Set $g_1 := \underline{w^2} - x\,z$, $g_2 := \underline{w\,y} - z^2$ and $g_3 := \underline{w\,z} - x\,y$. It follows that

$$\begin{aligned}
\operatorname{spoly}(g_1, g_2) &= y\,g_1 - w\,g_2 = \underline{w\,z^2} - x\,y\,z = z\,g_3\,, \\
\operatorname{spoly}(g_1, g_3) &= z\,g_1 - w\,g_3 = \underline{w\,x\,y} - x\,z^2 = x\,g_2 \\
\operatorname{spoly}(g_2, g_3) &= z\,g_2 - y\,g_3 = \underline{x\,y^2} - x^3\,.
\end{aligned}$$

Set $g_4 := \underline{x\,y^2} - x^3$. Since we also have

$$\begin{aligned}
\operatorname{spoly}(g_1, g_4) &= x\,y^2\,g_1 - w^2\,g_4 = \underline{w^2\,z^3} - x^2\,y^2\,z = z^3\,g_1 - x\,z\,g_4 \\
\operatorname{spoly}(g_2, g_4) &= x\,y\,g_2 - w\,g_4 = \underline{w\,z^3} - x\,y\,z^2 = z^2\,g_3 \\
\operatorname{spoly}(g_3, g_4) &= x\,y^2\,g_3 - w\,z\,g_4 = \underline{w\,z^4} - x^2\,y^3 = z^3\,g_3 - x\,y\,g_4\,,
\end{aligned}$$

the Buchberger criterion shows that g_1, g_2, g_3, g_4 is the reduced Gröbner basis of I with respect to the lexicographic order where $w > x > y > z$.

iii. We use *Macaulay2* [M2] to compute the leading term ideal for various weighted reverse lexicographic orders.

```
Macaulay2, version 1.24.11
with packages: ConwayPolynomials, Elimination, IntegralClosure, InverseSystems,
               Isomorphism, LLLBases, MinimalPrimes, OnlineLookup,
               PackageCitations, Polyhedra, PrimaryDecomposition, ReesAlgebra,
               Saturation, TangentCone, Truncations, Varieties

i1 : weightedLeadTerms = (a, b, c, d) -> (
         S = QQ[w,x,y,z, MonomialOrder => {Weights => {a,b,c,d}}];
         I = ideal(x*y-w*z, w*y-z^2, w^2-x*z);
         ideal leadTerm I);
```


CC BY NC ND

```
i2 : weightedLeadTerms(3, 1, 1, 1)

               2               2
o2 = ideal (x*y , w*z, w*y, w )

o2 : Ideal of S

i3 : weightedLeadTerms(3, 1, 1, 3)

             2        2
o3 = ideal (z , w*z, w )

o3 : Ideal of S

i4 : weightedLeadTerms(2, 3, 1, 3)

                  2        2
o4 = ideal (w*z, z , x*z, x y)

o4 : Ideal of S

i5 : weightedLeadTerms(1, 3, 3, 1)

o5 = ideal (x*z, w*y, x*y)

o5 : Ideal of S

i6 : weightedLeadTerms(3, 3, 1, 2)

                       2   3
o6 = ideal (w*y, w*z, w , z )

o6 : Ideal of S

i7 : weightedLeadTerms(3, 2, 1, 2)

                       2   3
o7 = ideal (w*y, w*z, w , z )

o7 : Ideal of S

i8 : weightedLeadTerms(1, 1, 1, 1)

                       2
o8 = ideal (x*y, w*y, w )

o8 : Ideal of S

i9 : weightedLeadTerms(2, 3, 3, 3)

             2
o9 = ideal (z , x*z, x*y)

o9 : Ideal of S

i10 : needsPackage "gfanInterface";

i11 : L = gfan I;
```


CC BY NC ND

```
i12 : netList table(#L // 2, 2, (j,k) -> ideal L#(2*j+k)#0)

      +-------------------------+----------------------+
      |                2     2  |            2       3 |
o12 = |ideal (w*z, w*y, w , x*y )|ideal (w*y, w , w*z, z )|
      +-------------------------+----------------------+
      |         2         2     |        3       2     |
      |ideal (w , w*z, z )      |ideal (w , w*z, z , x*z)|
      +-------------------------+----------------------+
      |              2       2  |        2             |
      |ideal (w*z, z , x*z, x y) |ideal (w , w*y, x*y)  |
      +-------------------------+----------------------+
      |                         |                 2    |
      |ideal (w*y, x*y, x*z)    |ideal (x*y, x*z, z )  |
      +-------------------------+----------------------+
```

□

P3.2. Fix the lexicographic order on the ring $S := \mathbb{K}[x_1, x_2, \dots, x_n]$ where $x_1 > x_2 > \cdots > x_n$. Let $\mathbf{A} := [a_{j,k}]$ be an $(m \times n)$-matrix with entries in the field $\mathbb{K}$. For all $1 \leqslant j \leqslant m$, let $f_j := a_{j,1}\, x_1 + a_{j,1}\, x_2 + \cdots + a_{j,n}\, x_n$ be the linear polynomial determined by the jth row of the matrix $\mathbf{A}$. Suppose that $\mathbf{B}$ is the reduced row echelon matrix associated to $\mathbf{A}$ and let $g_1, g_2, \dots, g_r$ be the linear polynomials determined by the nonzero rows in $\mathbf{B}$.

i. Prove that $\langle f_1, f_2, \dots, f_m \rangle = \langle g_1, g_2, \dots, g_r \rangle$.

ii. Show that $g_1, g_2, \dots, g_r$ form a Gröbner basis of the ideal $\langle f_1, f_2, \dots, f_m \rangle$.

iii. Explain why $g_1, g_2, \dots, g_r$ is the reduced Gröbner basis for the ideal $\langle f_1, f_2, \dots, f_m \rangle$.

Solution.

i. The reduced row echelon matrix $\mathbf{B}$ is obtained via a sequence of elementary row operations on the matrix $\mathbf{A}$. Thus, it suffices to show that, when $\mathbf{B}$ is obtained from $\mathbf{A}$ by an elementary row operation, the ideals generated by the linear polynomials determined by the nonzero rows are equal. We consider the three types of elementary row operations separately.

(*row add*) Suppose that the matrix $\mathbf{B}$ is obtained from the matrix $\mathbf{A}$ by replacing ith row in $\mathbf{A}$ with $\lambda \in \mathbb{K}$ times the jth row of $\mathbf{A}$ plus the ith row of $\mathbf{A}$. It follows that $g_i = \lambda\, f_j + f_i$, $f_i = g_i - \lambda\, g_j$, and $f_k = g_k$ for all $k \neq i$. We deduce that $\langle f_1, f_2, \dots, f_m \rangle = \langle g_1, g_2, \dots, g_r \rangle$.

(*row swap*) Suppose that the matrix $\mathbf{B}$ is obtained from the matrix $\mathbf{A}$ by interchanging ith and jth rows of $\mathbf{A}$ and leaving the other rows unchanged. It follows that $f_i = g_j$, $f_j = g_i$, and $f_k = g_k$ for all $k \notin \{i, j\}$. We see that $\langle f_1, f_2, \dots, f_m \rangle = \langle g_1, g_2, \dots, g_r \rangle$.

(*row multiple*) Suppose that the matrix $\mathbf{B}$ is obtained from the matrix $\mathbf{A}$ by multiplying the ith row of $\mathbf{A}$ by a nonzero element $\lambda \in \mathbb{K}$ and leaving the other rows unchanged. It follows that $f_i = \lambda\, g_i$ and $f_k = g_k$ for all $k \neq i$. Since $\lambda \neq 0$, we see that $g_i = \lambda^{-1}\, f_i$, so we deduce that $\langle f_1, f_2, \dots, f_m \rangle = \langle g_1, g_2, \dots, g_r \rangle$.

ii. As the matrix $\mathbf{B}$ is in reduced row echelon form, the set $\{\mathrm{LT}(g_1), \mathrm{LT}(g_2), \dots, \mathrm{LT}(g_r)\}$ consists of the leading (or pivot) variables. The number of pivot variables equals the number of nonzero rows in $\mathbf{B}$ (which also equals the rank of $\mathbf{A}$). As the leading monomials are pairwise relatively prime, the second Buchberger Criterion establishes that the linear polynomials $g_1, g_2, \dots, g_r$ are a Gröbner basis for the ideal $\langle g_1, g_2, \dots, g_r \rangle = \langle f_1, f_2, \dots, f_m \rangle$.

cc BY NC ND

iii. In reduced row echelon form, the leading entry in each nonzero row is 1 and each leading entry is the only nonzero entry in its column. In other words, we have $\mathrm{LC}(g_j) = 1$ for all $1 \leqslant j \leqslant r$ and, among all the linear polynomials $g_1, g_2, \dots, g_r$, the leading variable $\mathrm{LT}(g_j)$ only appears only in g_j. Hence, none of the monomials in $g_j - \mathrm{LT}(g_j)$ are divisible by an element of $\{\mathrm{LT}(g_1), \mathrm{LT}(g_2), \dots, \mathrm{LT}(g_r)\}$. Therefore, the polynomials $g_1, g_2, \dots, g_r$ are a reduced Gröbner basis. □

P3.3. Suppose that the numbers a, b, c, and d satisfy the equations

$$a+b+c+d = 3\,, \qquad a^3+b^3+c^3+d^3 = 237\,,$$
$$a^2+b^2+c^2+d^2 = 87\,, \qquad a^4+b^4+c^4+d^4 = 3\,123\,.$$

i. Prove that $a^5+b^5+c^5+d^5 = 13\,893$.
ii. Show that $a^6+b^6+c^6+d^6 \neq 17$.
iii. What are $a^6+b^6+c^6+d^6$ and $a^7+b^7+c^7+d^7$?

Solution.

i. To prove that $a^5+b^5+c^5+d^5 = 13\,893$, we show that $a^5+b^5+c^5+d^5 - 13\,893$ belongs to the ideal

$$I := \left\langle a+b+c+d-3, a^2+b^2+c^2+d^2-87, a^3+b^3+c^3+d^3-237, a^4+b^4+c^4+d^4-3123 \right\rangle\,.$$

More precisely, we use *Macaulay2* [M2] to compute a Gröbner basis for this ideal. If the remainder of $a^5+b^5+c^5+d^5 - 13\,893$ on division by the Gröbner basis is zero, then this element belongs to the ideal.

```
Macaulay2, version 1.24.11
with packages: ConwayPolynomials, Elimination, IntegralClosure, InverseSystems,
               Isomorphism, LLLBases, MinimalPrimes, OnlineLookup,
               PackageCitations, Polyhedra, PrimaryDecomposition, ReesAlgebra,
               Saturation, TangentCone, Truncations, Varieties

i1 : S = QQ[a,b,c,d];

i2 : I = ideal(a+b+c+d-3, a^2+b^2+c^2+d^2-87, a^3+b^3+c^3+d^3-237,
               a^4+b^4+c^4+d^4-3123);

o2 : Ideal of S

i3 : netList I_*

     +-------------------------+
o3 = |a + b + c + d - 3        |
     +-------------------------+
     | 2    2    2    2        |
     |a  + b  + c  + d  - 87   |
     +-------------------------+
     | 3    3    3    3        |
     |a  + b  + c  + d  - 237  |
     +-------------------------+
     | 4    4    4    4        |
     |a  + b  + c  + d  - 3123 |
     +-------------------------+
```


CC BY NC ND

```
i4 : transpose gens gb I

o4 = {-1} | a+b+c+d-3                                   |
     {-2} | b2+bc+c2+bd+cd+d2-3b-3c-3d-39               |
     {-3} | c3+c2d+cd2+d3-3c2-3cd-3d2-39c-39d+47 |
     {-4} | d4-3d3-39d2+47d+210                         |

              4      1
o4 : Matrix S  <-- S
```

To reduce $a^5 + b^5 + c^5 + d^5 - 13\,893$ with respect to the ideal I, *Macaulay2* [M2] automatically computes a (partial) Gröbner basis for I.

```
i5 : (a^5+b^5+c^5+d^5-13893) % I == 0

o5 = true
```

ii. We establish that $a^6 + b^6 + c^6 + d^6 - 17$ does not belong to the ideal I. We use *Macaulay2* [M2] to show the remainder of $a^6 + b^6 + c^6 + d^6 - 17$ on division by the Gröbner basis is not zero.

```
i6 : (a^6+b^6+c^6+d^6-17) % I != 0

o6 = true
```

iii. To determine the values of $a^6 + b^6 + c^6 + d^6$ and $a^7 + b^7 + c^7 + d^7$, we compute their remainders modulo the Gröbner basis.

```
i7 : (a^6+b^6+c^6+d^6) % I
o7 = 134067

o7 : S

i8 : (a^6+b^6+c^6+d^6-134067) % I == 0

o8 = true

i9 : (a^7+b^7+c^7+d^7) % I

o9 = 747477

o9 : S

i10 : (a^7+b^7+c^7+d^7 - 747477) % I == 0

o10 = true
```

Therefore, we have $a^6 + b^6 + c^6 + d^6 = 134\,067$ and $a^7 + b^7 + c^7 + d^7 = 747\,477$. □

P3.4. The ***Whitney umbrella*** is the image of the polynomial map $\rho\colon \mathbb{A}^2 \to \mathbb{A}^3$ defined by

$$(u, v) \mapsto (uv, v, u^2)\,.$$

i. Find the equation(s) for the smallest affine subvariety in $\mathbb{A}^3$ containing the Whitney umbrella.

ii. Show that the parametrization fills up this affine subvariety over $\mathbb{C}$ but not over $\mathbb{R}$. Over $\mathbb{R}$, exactly what points are omitted?

CC BY NC ND

iii. Demonstrate that the parameters u and v are not always uniquely determined by a point in $\mathbb{A}^3$. Find the points where uniqueness fails.

Solution.

i. The polynomial implicitization theorem implies that the smallest affine subvariety corresponds to the elimination ideal. Thus, we find the equation for the Zariski closure of the Whitney umbrella in *Macaulay2* [M2] as follows.

```
Macaulay2, version 1.24.11
with packages: ConwayPolynomials, Elimination, IntegralClosure, InverseSystems,
               Isomorphism, LLLBases, MinimalPrimes, OnlineLookup,
               PackageCitations, Polyhedra, PrimaryDecomposition, ReesAlgebra,
               Saturation, TangentCone, Truncations, Varieties

i1 : S = QQ[x,y,z,u,v];

i2 : I = ideal(x - u*v, y-v, z-u^2)

                                      2
o2 = ideal (- u*v + x, y - v, - u  + z)

o2 : Ideal of S

i3 : eliminate(I, {u, v})

             2       2
o3 = ideal(y z - x )

o3 : Ideal of S
```

The smallest affine subvariety containing the Whitney umbrella is $\mathrm{V}(y^2 z - x^2) \subset \mathbb{A}^3$.

ii. Suppose that $(a, b, c) \in \mathbb{A}^3(\mathbb{C})$ is a point such that $b^2 c - a^2 = 0$. Set $v := b$. When $b = 0$, we have $0 = b^2 c = a^2$, so $a = 0$. In this case, let u be either square root of c. When $b \neq 0$, set $u := a\, b^{-1}$. Since $b^2 c = a^2$ we have $c = a^2 b^{-2} = u^2$. Therefore, the parametrization fills up the algebraic set over $\mathbb{C}$.

Suppose that $(a, b, c) \in \mathbb{A}^3(\mathbb{R})$ is a point such that $b^2 c - a^2 = 0$. Set $v := b$. When $b \neq 0$, set $u := a\, b^{-1}$. Since $b^2 c = a^2$, we have $c = a^2 b^{-2} = u^2$. When $b = 0$, we have $0 = b^2 c = a^2$, so $a = 0$. In this case, u can be either square root of c if and only if $c \geqslant 0$. Hence, the parametrization over $\mathbb{R}$ omits the points $(0, 0, c)$ with $c < 0$.

iii. Suppose that we have $x = u_1 v_1 = u_2 v_2$, $y = v_1 = v_2$ and $z = u_1^2 = u_2^2$. It follows that $v_1 = v_2$, $v_1(u_1 - u_2) = 0$ and $(u_1 - u_2)(u_1 + u_2) = 0$. When $v_1 \neq 0$, we have $u_1 = u_2$. When $v_1 = 0$, we have $u_1 = \pm u_2$ and we do not have unique values for the parameters. Thus, the parameters are not uniquely determined by a point on the Whitney umbrella if and only if the point has the form $(0, 0, z)$. □

P3.5. Consider the ideal $I := \langle 9x^2 - xy - 180, 81x^2 - 20xy - y^2 - 1620 \rangle$ in $\mathbb{Q}[x, y]$.

i. Find Gröbner basis for $I \cap \mathbb{Q}[x]$ and $I \cap \mathbb{Q}[y]$.

ii. Find all solutions to the equations $9x^2 - xy = 180$ and $81x^2 - 20xy - y^2 = 1620$ in $\mathbb{C}$.

iii. Which of the solutions are rational?

iv. What is the smallest field $\mathbb{K}$ containing $\mathbb{Q}$ such that all solutions lie in $\mathbb{K}$?

Solution.

i. We compute the elimination ideals in *Macaulay2* [M2] as follows.

```
Macaulay2, version 1.24.11
with packages: ConwayPolynomials, Elimination, IntegralClosure, InverseSystems,
               Isomorphism, LLLBases, MinimalPrimes, OnlineLookup,
               PackageCitations, Polyhedra, PrimaryDecomposition, ReesAlgebra,
               Saturation, TangentCone, Truncations, Varieties

i1 : S = QQ[y,x, MonomialOrder => Lex];

i2 :  I = ideal(9*x^2-x*y-180, 81*x^2-20*x*y-y^2-1620)

                2               2            2
o2 = ideal (9x  - x*y - 180, 81x  - 20x*y - y  - 1620)

o2 : Ideal of S

i3 : Iy = eliminate(I, x)

              3
o3 = ideal(y  - 1089y)

o3 : Ideal of S

i4 : Ix = eliminate(I, y)

              4      2
o4 = ideal(x  - 29x  + 180)

o4 : Ideal of S
```

We deduce that $I \cap \mathbb{Q}[x] = \langle x^4 - 29\,x^2 + 180 \rangle$ and $I \cap \mathbb{Q}[y] = \langle y^3 - 1089\,y \rangle$.

ii. To find all solutions, we first factor:

```
i5 : factor Iy_0

o5 = (y)(y - 33)(y + 33)

o5 : Expression of class Product

i6 : factor Ix_0

                     2
o6 = (x - 3)(x + 3)(x  - 20)

o6 : Expression of class Product
```

Therefore, we have $x \in \{-3, 3, -2\sqrt{5}, 2\sqrt{5}\}$. To find the corresponding y-values, observe that

```
i7 : gens gb I

o7 = | x4-29x2+180 y-x3+20x |

               1       2
o7 : Matrix S  <--- S
```


CC BY NC ND

which implies that $y = x^3 - 20x$. Therefore, the complete set of solution over $\mathbb{C}$ is $\{(-3, 33), (3, -33), (-2\sqrt{5}, 0), (2\sqrt{5}, 0)\}$.

iii. The rational solutions are $\mathbb{V}(I) \cap \mathbb{A}^2(\mathbb{Q}) = \{(-3, 33), (3, -33)\}$.

iv. The smallest subfield of $\mathbb{C}$ containing all the solutions is $\mathbb{Q}(\sqrt{5})$. $\square$

References

[M2] The *Macaulay2* project authors, *Macaulay2, a software system for research in algebraic geometry*, 2024. available at https://macaulay2.com.